EXAM FUNCTIONAL PROGRAMMING
Thursday the 6th of November 2014, 13.30 h. - 16.30 h.

Name: Student number:

Before you begin: Do not forget to write down your name and student number above. If necessary, explain your answers (in English or Dutch). For multiple choice questions, clearly circle what you think is the (one and only) best answer. Use the empty boxes under the other questions to write your answer and explanations in. Use the empty paper provided with this exam only as scratch paper (kladpapier). At the end of the exam, only hand in the filled-in exam paper. Answers will not only be judged for correctness, but also for clarity and conciseness. A total of one hundred points can be obtained; divide by 10 to obtain your grade. Good luck!

In any of your answers below you may (but do not have to) use the following well-known Haskell functions/operators: *id*, *concat*, *foldr* (and variants), *map*, *filter*, *const*, *flip*, *fst*, *snd*, *not*, (.), *elem*, *take*, *drop*, *takeWhile*, *dropWhile*, *head*, *tail*, (++), *lookup* and all members of the type classes *Eq*, *Num*, *Ord*, *Show* and *Read*.

1. (i) Define a type class *Finite a* (eindig), that has one member *values* that enumerates all (finitely many) values of type a.

.../5

```
class Finite a where
  values :: [a]
```

(ii) Define a suitable instance for *Finite Bool*.

.../3

```
instance Finite Bool where
  values = [False, True]
```

(iii) Define a suitable instance for *Finite* (a, b, c) with a list comprehension, for the case that a, b and c are instances of *Finite*.

.../6

```
instance (Finite a, Finite b, Finite c) => Finite (a, b, c) where
  values = [(x, y, z) | x <- values, y <- values, z <- values]
```

(iv) Why is it not possible to add a member *size* :: *Int* (that returns the length of *values*) to the *Finite* type class?

.../5 Because then the type inferencer cannot tell which instance you want to use, since the a of *Finite a* is not visible in the type of *size*.

2. In this question we deal with a function *segs* :: [*a*] -> [[*a*]] which returns all the segments of the argument list. A list *L1* is a segment of another list *L2*, if you can obtain *L1* from *L2* by dropping any number of elements (including 0) at the beginning of *L2*, and dropping any number of elements (including 0) at the end of *L2*.

   (i) What are the segments of [1,2,3,4]?

   . . . /**4** [[],[4],[3],[3,4],[2],[2,3],[2,3,4],[1],[1,2],[1,2,3],[1,2,3,4]]

   (ii) Explain how you can compute *segs* (*x* : *xs*) from *segs xs* (for example by using concrete values for *x* and *xs*)

   . . . /**6** Segments of *segs xs* are also in *segs* (*x* : *xs*). What we are missing are the segments that start with *x*. Those are exactly the *inits* (*x* : *xs*) but with the empty list removed (since we have that one already).

   (iii) Now, write the function *segs* :: [*a*] -> [[*a*]]

   . . . /**6** This is from the reader:

$$\begin{aligned} inits'\ [\,] &= [[\,]] \\ inits'\ (x : xs) &= [\,] : map\ (x:)\ (inits'\ xs) \end{aligned}$$

$$\begin{aligned} segs\ [\,] &= [[\,]] \\ segs\ (x : xs) &= segs\ xs\ ++\ map\ (x:)\ (inits'\ xs) \end{aligned}$$

   There is an alternative for the last line:

$$segs\ (x : xs) = segs\ xs\ ++\ tail\ (inits\ (x : xs))$$

(iv) Write a QuickCheck property *numberProp* :: [*Int*] -> *Property* that tests whether *segs xs* has the correct number of segments, but only for input lists of length at least 3.

. . . /**6**

```
numberProp :: [Int] -> Property
numberProp xs = lenxs >= 3 ==> length (segs xs) == nrOfSegs lenxs
  where
    lenxs = length xs
    nrOfSegs 0 = 1
    nrOfSegs n = nrOfSegs (n - 1) + n
```

Alternatively, you can also have seen that nrOfSegs n = (n+1)*n ‘div‘ 2 and use that instead.

3. Given is the following datatype for trees:

```
data Tree = Leaf | Bin Tree Int Tree deriving Eq
```

(i) Define a function *listLike* :: *Tree* -> *Bool* that returns *True* if every *Bin* node has at most one non-Leaf child.

. . . /**7**

```
listLike :: Tree -> Bool
listLike Leaf     = True
listLike (Bin l _ r) = (r == Leaf && listLike l) || (l == Leaf && listLike r)

  -- or the somewhat less cryptic

listLike Leaf                = True
listLike (Bin Leaf _ r)      = listLike r
listLike (Bin l _    Leaf)   = listLike l
listLike (Bin _ _    _)      = False
```

(ii) Assume that an instance *Arbitrary Tree* has been defined, write a generator *genNLLTree* :: *Gen Tree* for arbitrary trees that are *not* list-like.

. . . /**7** This one does the trick:

```
genNLLTree :: Gen Tree
genNLLTree =
  do
    ts <- sequence [arbitrary, arbitrary, arbitrary, arbitrary]   -- start with 3 trees
    let nlls = filter (not . listLike) ts
    if null nlls then    -- all are listLike
      do
        i1 <- arbitrary
        i2 <- arbitrary
        i3 <- arbitrary
        return (Bin (Bin (ts !! 0) i3 (ts !! 1)) i1
          (Bin (ts !! 2) i2 (ts !! 3)))
    else
      return (head nlls)
```

Another acceptable alternative is the following straightforward solution:

```
genNLLTree :: Gen Tree
genNLLTree =
```

4. Given are the following definitions (with line numbers):

(1) $id\ x = x$

(2) $flip\ f\ x\ y = f\ y\ x$

(3) $reverse\ [\,] = [\,]$
(4) $reverse\ (x : xs) = reverse\ xs\ ++\ [x]$

(5) $foldr\ f\ e\ [\,] = e$
(6) $foldr\ f\ e\ (x : xs) = f\ x\ (foldr\ f\ e\ xs)$

(7) $foldl\ f\ e\ [\,] = e$
(8) $foldl\ f\ e\ (x : xs) = foldl\ f\ (f\ e\ x)\ xs$

(i) Prove by induction that $foldr\ (:)\ [\,] = id$ (use the line numbers above when you refer to a particular given equation in your proof):

. . . /**7** Bewijs met inductie naar xs:

	`foldr (:) []`	`id`

`IH xs`	`foldr (:) [] xs`	`id xs`
		= (def. `id (1)`)
		`xs`

`[]`	`foldr (:) [] []`
	= (def. `foldr (5)`)
	`[]`
		`[]`

`x:xs`	`foldr (:) [] (x:xs)`
	= (def. `foldr (6)`)
	`(:) x (foldr (:) [] xs)`
	= (IH xs)
	`(:) x xs`
	= (postfix to infix)
	`x : xs`
		`x : xs`

(ii) Prove by induction that *foldr f e* (*reverse xs*) = *foldl* (*flip f*) *e xs* for all f, e and xs of the right type. You may use (without proof) the following lemma: *foldr f e* (as ++ [b]) = *foldr f* (f b e) as for all suitable f, e, as and b (again, use the line numbers when you refer to a particular given equation in your proof).

. . . /**13** Bewijs met inductie naar xs:

IH xs	`foldr f e (reverse xs)`	`foldl (flip f) e xs`

`[]`	`foldr f e (reverse [])`
	= (def. `reverse (3)`)
	`foldr f e []`
	= (def. `foldr (5)`)
	`e`
		`foldl (flip f) e []`
		= (def. `foldl (5)`)
		`e`

`x:xs`	`foldr f e (reverse (x:xs))`
	= (def. `reverse (4)`)
	`foldr f e (reverse xs++[x])`
	= (hulpwet hierboven)
	`foldr f (f x e) (reverse xs)`
		`foldl (flip f) e (x:xs)`
		= (def. `foldl (8)`)
		`foldl (flip f) (flip f e x) xs`
		= (def. `flip (2)`)
		`foldl (flip f) (f x e) xs`
		= (IH met `f x e` voor `e`)
		`foldr f (f x e) (reverse xs)`

5. **. . . /25. Correct answers are: b, d, d, c, c** The following multiple choice questions are each worth 5 points.

(i) Which of the following is true?

a. The function *return* is idempotent (i.e. *return* (*return a*) can safely be replaced by *return a*).

b. There exist expressions of type *IO* (*IO Int*).

c. If you define an instance of the class *Eq* you have at least to specify the function (==).

d. The class *Enum* has a fixed number of instances.

(ii) I A jargon is a special kind of domain-specific language.

II It is easier to achieve fluency with a deeply embedded DSL than with a shallowly embedded DSLs.

a. Both I and II are true

b. Only I is true

c. Only II is true

d. Both I and II are false

(iii) Which observation is correct when comparing the types of (*map map*) *map* and *map* (*map map*)?

a. The type of the first is less polymorphic than the type of the second.

b. The type of the first is more polymorphic than the type of the second.

c. The types are the same, since function composition is associative.

d. One of the expressions is type incorrect.

(iv) What is the type of *foldr flip*?

a. $b \to (b \to a \to b) \to [a] \to b$

b. $(a \to b) \to [b \to b] \to a \to b$

c. $(a \to b) \to [a \to (a \to b) \to b] \to a \to b$

d. The expression is type incorrect

(v) In the Haskell prelude the list constructor [] has been made an instance of the class *Monad*:

```
instance Monad [] where
  ma >>= a2mb = concat (map a2mb ma)
  return a = [a]
```

Which of the following equals [*f x y* | *x* <- *expr1*, *y* <- *expr2*]?

a.
```
do return (f x y)
   where do x <- expr1
            y <- expr2
```

b.
```
do x <- expr1
   y <- expr2
   f x y
```

c.
```
do x <- expr1
   y <- expr2
   return (f x y)
```

d.
```
do y <- expr2
   x <- expr1
   return (f x y)
```